
BY J IM MOORE 
Working Smarter

Static vs. Dynamic Linkage
in COBOL

Every so often, I encounter programmers
(usually COBOL programmers) who

don't understand the differences between
dynamic and static linkage in the OS/390
environment. So, this month I will discuss a
short and simple program that examines the
CDE (Content Directory Entry) chain. The
program navigates the CDE linked list that
is anchored in the primary TCB (Task
Control Block) within an address space.

DYNAMIC VS. STATIC

Since "dynamic vs. static" seems to be
the source of a lot of confusion for COBOL
programmers, it is only fair to clarify the
differences right away. Assembler pro-
grammers are usually pretty clear on the
differences because they are coding their
programs to be either dynamic or static
(LINK/LOAD/XCTL as opposed to CALL). 

In COBOL, the compile-time option
DYNAM / NODYNAM controls how pro-
grammer-coded CALL statements (or
CICS/IDMS pre-processor inserted CALL
statements) are converted by the compiler.
With DYNAM active at compile-time,
CALL statements essentially become
LOADs (SVC 06) followed by a branch to
the entry point address. No V-Type constants
are created in the object deck's ESD when
DYNAM is active at compile-time. Therefore,
a DYNAM compiled COBOL program will
rarely encounter "UNRESOLVED EXTER-
NAL REFERENCES" at link-time. 

Incidentally, the obsolete RES / NORES
COBOL compile-time option also was used
to control dynamic calls. The difference was
that RES / NORES applied to the compiler
run-time routines, not programmer-coded
CALLs. RES / NORES is no longer a valid
compile-time option under Language
Environment (LE) COBOL.

NODYNAM, as a compile-time option
turns a COBOL program's CALL statements
into Assembler-style CALLs. V-Type
constants are created that must be resolved
by the binder/linker. NODYNAM load
modules must have these external addresses
statically imbedded. That is, the external
CSECTS must be resolved and packaged
into the run-unit at link-time. Essentially, a
NODYNAM CALL in COBOL looks like
the following in Assembler:

L     R15,VCON
BASR  R14,R15
.
.
VCON DC V(EXTRNPGM)

An all too common occurrence on
OS/390 systems is a COBOL program
abending with an S0C1. An S0C1 results
from trying to execute something that isn't
an instruction. This can happen if a pro-
grammer ignores or misses the "unresolved
external reference" from a link-edit. 

Upon program execution, when the code
gets to the "L R15,VCON - BASR
R14,R15", a fullword of binary zeros is
loaded into R15 (the unresolved V-Con).
The machine obediently branches to
address zero. There is not a valid instruction
at this address. The result is an S0C1.

Another thing to be aware of is the
COBOL "CALL identifier-1" format. This
format of the CALL statement is always
dynamic, regardless of the DYNAM /
NODYNAM setting at compile-time. Using
this format of the COBOL CALL statement
is one way to ensure that a COBOL program's
calls are always dynamic regardless of
compile-time parameter settings.

WHICH IS BETTER, STATIC
OR DYNAMIC?

Personally, I prefer dynamic; however,
many people prefer static. Both methods
have advantages and caveats. The following
presents a brief comparison.

Dynamic (DYNAM or LINK/LOAD)
The pros of dynamic include the following:

● smaller load modules
● easier to implement maintenance to

heavily called subroutines
● no re-linking of mainline routines

when subroutines change
● under LE most compiler run-times are

always dynamic

Following are the cons of dynamic:

● programs loaded at run-time-potentially,
more I/O to JOBLIB

● missing program can cause S806.

Static (NODYNAM or V-Type Constant
in ESD)

Following are the pros of static:

● Entire program with all addresses
resolved loaded into memory

● Load modules are self-contained and
can run without elaborate search paths

Personally, I prefer
dynamic; however,
many people prefer
static. Both methods

have advantages
and caveats.

©2000 Technical Enterprises, Inc. Reproduction of this document without permission is prohibited.



TECHNICAL SUPPORT • JUNE 2000

The cons of static are as follows:

● larger load modules
● subroutines duplicated in many load

modules
● re-linking required to pick up new

versions of subroutines
● NORES gone from COBOL so  run-

times cannot be statically imbedded

THINGS TO AVOID

COBOL programmers should avoid the
following two techniques:

1. Use of the CALL/CANCEL matched
set in conjunction with DYNAM
compiled COBOL programs. 

2. Use of binder INCLUDE statements at
SYSLIN for DYNAM compiled
COBOL programs.

The "CALL 'pgm' CANCEL" statement
shouldn't be used in modern COBOL
environments. The CANCEL acts as a CDE
destructor, so that the next time the program
is called a fresh copy is loaded. This technique
was used to ensure that a called subroutine
was always entered in an initial state but it
caused excessive I/O to JOBLIB.

A much better method of accomplishing
this nowadays is to give the program the initial
attribute with the "IS INITIAL" clause of
the PROGRAM-ID:

PROGRAM-ID. COBPROG IS INITIAL.

This causes the compiler to include extra
code that will ensure that the program is
always entered in a "first-time called" state
with all VALUE clause initialization intact.

Use of the INCLUDE statement in con-
junction with an object deck that contains
SVC 06 invocations is useless. Using 

Continued on page 52.
Working Smarter continued from page 49.

superfluous INCLUDE statements won't
cause any link-time problems but doing so
is a waste of time and space. The binder will
slavishly include the named load module/
object deck but all that this is doing is making
the link-edited run-unit larger than it should
be. At run-time, the SVC 06 will cause a
dynamic load anyway. 

WALKING THE CDE CHAIN 

Figure 1 shows a simple LE COBOL
program that "walks" the linked list known

as the Content Directory Entry list (refer to
DSECT IHACDE in SYS1.MODGEN).
Initially, before it calls IEFBR14, it performs
the routine named "1000-Scan-CDE-

Chain." Then, it issues a call to IEFBR14
and walks the CDE chain again. 

The purpose of this program is to see
what new programs have been loaded into

000100 Identification Division.                                         
000200 Program-id. SCANCDES is Initial.                                 
000300 Author. Jim Moore - Concentrated Logic Incorporated.              
000400 Environment Division.                                            
000500 Data Division.                                                   
000600 Working-Storage Section.                                         
001100/ === All Control Blocks must reside in the Linkage Section ======
001200 Linkage Section.                                                 
001400 01 PSA-Block.                                                   
001500 05                      Pic  X(536).                         
001600 05  TCB-Pointer         Pointer.                             
001700 01 TCB-Block.                                                   
001800 05                      Pic  X(44).                          
002400 05  TCBJPQ-Field        Pic S9(08) Binary.                   
002500         88  Empty-JPQ       Value Zeros.                         
002600 05  TCBJPQ-Pointer      Redefines TCBJPQ-Field,              
002610                             Pointer.                             
002700 01 CDE-Block.                                                   
002800 05  Next-CDE            Pointer.                             
002900 05                      Redefines Next-CDE                   
003000                             Pic  X(04).                          
003100         88  Last-Module     Value Low-Values.                    
003200 05                      Pic  X(04).                          
003300 05  Module-Name         Pic  X(08).                          
003400                                                                  
003500/ == Chain PSA -> TCB -> to set up CDE walk                       
003600 Procedure Division.                                              
003700 Set Address of PSA-Block to Null.                           
003800 Set Address of TCB-Block to TCB-Pointer.                    
003810                                                                  
003900 Display "*-- Programs in CDE Chain Before 'DYNAM' CALL--*"   
003901 Perform 1000-Scan-CDE-Chain               Thru 1000-Exit.    
003902 CALL 'IEFBR14'.                                              
003903 Display "*-- Programs in CDE Chain After  'DYNAM' CALL--*"   
003904 Perform 1000-Scan-CDE-Chain               Thru 1000-Exit.    
003910 GOBACK.                                                      
003920                                                                  
003930/ == Walk CDE to Display Content Directory Entries                
004000 1000-Scan-CDE-Chain.                                             
004100 Set Address of CDE-Block to TCBJPQ-Pointer.                 
006000 If Empty-JPQ                                                 
006100        Display "*None*"                                          
006200 Else                                                         
006400 Perform Until Last-Module                                    
006600        Display "Content Directory (JPQ) Name is " Module-Name    
006700        Set Address of CDE-Block to Next-CDE                      
006800 End-Perform.                                                 
006801 1000-Exit. Exit.                                                 

FIGURE 1: A COBOL FOR MVS PROGRAM THAT WALKS THE CDE CHAIN

*-- PROGRAMS IN CDE CHAIN BEFORE 'DYNAM' CALL--*
CONTENT DIRECTORY (JPQ) NAME IS IGZCLNK
CONTENT DIRECTORY (JPQ) NAME IS IGZEINI
CONTENT DIRECTORY (JPQ) NAME IS IGZCPCO
CONTENT DIRECTORY (JPQ) NAME IS IGZCPAC
CONTENT DIRECTORY (JPQ) NAME IS IGZCEV5
CONTENT DIRECTORY (JPQ) NAME IS CEEEV005
CONTENT DIRECTORY (JPQ) NAME IS CEEPLPKA
CONTENT DIRECTORY (JPQ) NAME IS CEEBINIT
*-- PROGRAMS IN CDE CHAIN AFTER  'DYNAM' CALL--*
CONTENT DIRECTORY (JPQ) NAME IS IEFBR14         <-- NEW LOADS
CONTENT DIRECTORY (JPQ) NAME IS IGZCFCC         <-- NEW LOADS
CONTENT DIRECTORY (JPQ) NAME IS IGZCLNK
CONTENT DIRECTORY (JPQ) NAME IS IGZEINI
CONTENT DIRECTORY (JPQ) NAME IS IGZCPCO
CONTENT DIRECTORY (JPQ) NAME IS IGZCPAC
CONTENT DIRECTORY (JPQ) NAME IS IGZCEV5
CONTENT DIRECTORY (JPQ) NAME IS CEEEV005
CONTENT DIRECTORY (JPQ) NAME IS CEEPLPKA
CONTENT DIRECTORY (JPQ) NAME IS CEEBINIT

FIGURE 2: THE "DISPLAY" RESULTS OF A DYNAM RUN

memory. Under LE COBOL, the program
named IGZCFCC has been loaded as well
as IEFBR14 (see Figure 2). IGZCFCC is
the dynamic linkage manager of the
COBOL run-time. 

CONCLUSION

You must compile the program shown in
Figure 1 with DYNAM active at compile-
time in order to see the results shown in
Figure 2. An interesting thing to try (and I
have tried this) is to recompile the program
with NODYNAM and then run it again. You
will need SYS1.LINKLIB allocated to
SYSLIB in the binder step to resolve the
IEFBR14 V-Con. What would you expect
the results to be?  

